Introducing Statistical Graphics (SG):

Vancouver UG
May 2018

Mary Harding
SAS Canada
Education Department

Agenda

- Introduction to Statistical Graphics
- PROC SGPLOT
  - General purpose
  - Overlays
  - Statistical
- Quick look at SGPANEL and SGSCATTER
- Supporting Statements/Facilities
- Questions

Statistical Graphics (ODS Graphics)

ODS Graphics is based on Graph Template Language (GTL), which provides the power and flexibility to create many complex statistical and non-statistical graphs.







Prior to SAS 9.3, ODS Graphics was part of SAS/GRAPH. Starting in SAS 9.3, ODS Graphics is part of Base SAS.

r108d01

Statistical Graphics

Statistical Graphics, also referred to as *ODS Graphics*, as it is an extension of the SAS Output Delivery System (ODS).

ODS Graphics Procedures	Provide a concise syntax for creating statistical and non-statistical graphs.

ODS Graphics Procedures

The ODS Graphics procedures provide a traditional SAS procedure interface for the most commonly used features of the SAS Graph Template Language.

SGPLOT	Creates a single-celled graph, and multiple plots are overlaid within a single set of axes.
SGPANEL	Creates classification panels containing either a simple plot or multiple overlaid plots.
SGSCATTER	Creates panels with multiple scatter plots.

SGPLOT Procedure

The SGPLOT procedure can produce the following and more:

Basic Plots	Scatter plots, series plots, band plots, bubble plots, high-low charts, needle plots and vector plots
Fit and Confidence Plots	Loess curves, regression curves, penalized B-spline curves, and ellipses
Distribution Plots	Histograms, box plots, and density curves
Categorization Plots	Bar charts, dot plots, and bar-line charts

SGPLOT Procedure

Combining graphs: Which can be overlaid together?

Basic Plots → Overlay only with → **Other Basic Plots and Fit and Confidence Plots**

Distribution Plots → Overlay only with → **Other Distribution Plots**

Categorization Plots → Overlay only with → **Other Categorization Plots**

GENERAL PURPOSE GRAPHS

Bar Chart : Basic

```
proc sgplot data=orion.emps;
   hbar Country / response=Salary stat=mean;
run;
```



Bar Chart : Advanced

Add grid lines to Y-axis and assign label "Sales" to column Actual.

```
Yaxis grid
label='Sales';
```



Add Datalabels to bars, use Style attributes to adjust font weight, color and size

```
VBAR product /
response=actual
group=quarter
            stat=sum
datalabel
datalabelattrs=(weight
=bold color='dark
blue' size=10pt)
dataskin=gloss ;
```

Suppress the X axis column Product: `xaxis display=(nolabel);`

Assigns colours in specified order:

```
styleattrs datacolors=(orange blue green yellow);
```

Make it look slick with dataskin=gloss|sheen|crisp|Matte|pressed

10

Bar Chart : Advanced

```
Proc SGPLOT data=sashelp.prdsale;
 format actual dollar6.0;
 title 'Actual Sales by Product and Quarter';
 vbar product / response=actual group=quarter
      stat=sum
      datalabel datalabelattrs=(weight=bold
      color='dark blue' size=10pt)
      dataskin=gloss;
 xaxis display=(nolabel);
 yaxis grid  label='Sales';
 styleattrs datacolors=(orange blue green
 yellow);
run;
```

High Low with Background: Use Annotate



Use SQL or Data step to calculate Minimum and Maximum values of MSRP by Make.

PROC SGPLOT Options:

Suppress background and borders, use annotate dataset:

```
proc sgplot data=cars
nowall noborder
sganno=annoImage ;
```

Create Annotate dataset with background image image="&path\cars.jpg"; layer='back';

Assign other options to annotate dataset such as dataspace, which controls positioning and scaling.

This graph has DRAWSPACE= 'GRAPHPERCENT' as we want the entire annotation (background image to take up the entire graph

High Low with Background: Annotate Table

```
data annoImage;
 length function $10;
 function='image'; height=100; width=100;
 transparency=0.8; drawspace='GraphPercent';
 image="&path\cars.jpg"; layer='back';
run;

ods graphics / reset width=9in height=5in
imagename='Car Make Minimum & Maximum MSRP';
proc sql;
 create table cars as
 select make ,min(msrp) format=Dollar14.0
 as min_MSRP, max(msrp) as max_msrp
 format=Dollar14.0
  from sashelp.cars
    where origin in('Europe','USA')
      group by make;
quit;
```

High Low with Background :SGPLOT

```
title  color=bib height=16pt 'Minimum and Maximum
MSRP values by Make';
title2 color=bib 'Europe and USA';
PROC SGPLOT data=cars  nowall noborder
   sganno=annoImage ;
   highlow x=make low=min_MSRP high=max_msrp /
   type=bar  name='h' lineattrs=graphoutlines
   dataskin=gloss  fillattrs=(transparency=0
   color=orange);
 xaxis display=(nolabel noline noticks);
 yaxis display=(nolabel noline)grid
 offsetmax=0.1;
keylegend 'h' / noopaque noborder fillheight=14
fillaspect=golden;
RUN;
```

High Low with Background: Drawspace





drawspace='wallpercent';







drawspace='graphpercent';



SGPLOT: Series

Monthly Sales for France Italy and UK Orion Offices

Insert a text inset box within graph:

```
Inset "Monthly target is
$300,000"/position=topright
textattrs=(size=14pt color=white
weight=bold);
```

Show Distinct X axis numeric values

```
XAXIS type=discrete
labelattrs=(size=15pt
color=white);
```

Insert reference line with attributes

```
refline 300000/label='$300,000'
 lineattrs=( color=blue
pattern=dash thickness=0.5mm) ;
```

SGPLOT: Series

```
ods graphics / border reset width=9in height=5in ;
Title "Monthly Sales for France Italy and UK Orion
Offices";
proc sgplot data=retail.profit sganno=annoImage_vic nowall
where Country in ('France','Italy','UK')
       and Year=2007;
series x=Month y=Sales / group=Company
lineattrs=(thickness=0.15cm);
refline 300000/label='$300,000'
lineattrs=(color=blue pattern=dash thickness=0.5mm) ;
XAXIS type=discrete labelattrs=(size=15pt
color=white);/*show exact numeric values for month*/
YAXIS label='Monthly Sales' labelattrs=(size=15pt
color=white);
inset "Monthly target is $300,000"/position=topright
  textattrs=(size=14pt color=white weight=bold);
run;
```

OVERLAY GRAPHS

SGPLOT : Multiple Plots

Overlaid graphs are



Bar-Line Chart

```
proc sgplot data=orion.orders;
   vbar CustomerCountry /
        response=TotalRetailPrice;
   vline CustomerCountry /
         response=Quantity y2axis;
run;
```



HIGHLOW and SCATTER Overlay with Segment Labels



HIGHLOW and SCATTER Overlay with Segment Labels

Data preparation:

```
data HighLow;
  format low mid high actualsum dollar7.0;
  format highlabel dollar8.0;
  retain low 0;
  set prdsale (drop=_type_);
  by product;

  if first.product then low=0;
  high=low+ActualSum;
  mid = (low+high)/2;

  if last.product then highlabel=high;
  output;
  low=high;
run;
```

HIGHLOW and SCATTER Overlay with Segment Labels

```
/*--Stacked Group with segment Labels--*/
ods graphics / reset width=8in height=5in;
proc sgplot data=HighLow;
  title 'Actual Sales by Product and Quarter';
  highlow x=product low=low high=high / group=quarter
          type=bar lineattrs=(pattern=solid) dataskin=gloss;
  scatter x=product y=mid / markerchar=actualsum
  markercharattrs=(size=12pt);
  xaxis display=(nolabel);
  yaxis grid offsetmin=0 label='Sales';
run;
```

STATISTICAL

Histogram and Density Curve

```
proc sgplot data=orion.emps;
    histogram Salary;
    density Salary;
run;
```



Loess and Scatter Plot Overlaid

```
proc sgpanel data=sashelp.cars;
where origin in ("Asia" "Europe");
panelby origin;
scatter x=mpg_city y=msrp / group=type;
loess x=mpg_city y=msrp / nomarkers name="fit";
run;
```



QUICK LOOK AT SGPANEL AND SGSCATTER

SGPANEL Procedure

The SGPANEL procedure creates a panel for the values of one or more classification variables.




```
proc sgpanel data=orion.profit;
    panelby Company;
    series x=YYMM y=Sales;
run;
```

r108d03

SGSCATTER Procedure

The SGSCATTER procedure creates panels with multiple scatter plots.

The procedure syntax supports the following features:

- three types of graph layouts: PLOT, COMPARE, and MATRIX
- basic scatter plots, fit and confidence plots, and distribution plots

MATRIX Layout

The MATRIX statement creates scatter plots where each variable is graphed against each other.

```
proc sgscatter data=orion.profit;
   where Country in ('France','Italy','UK');
   matrix sales salaries profit / group=company
          diagonal=(histogram normal);
run;
```



SUPPORTING STATEMENTS/FACILITIES

Supporting Statements and Facilities

Further control and enhance Graphics

- AXIS Statements (including secondary axes)
- Legend Statements
- Reference Statements
- ODS graphics statement
- Annotate Facility

  more……

If your ODS graphs are not working, ODS Graphics may be off. On most operating systems the default is ON.

```
ODS Graphics on;
```

Supporting Statements and Facilities

Use supporting SG Statements to further control and enhance the plots.

AXIS Statements:

- Axis labels
- Grid lines
- Major and minor tick marks



Example

```
xaxis values=('US' 'GB' 'FR' 'ES')
fitpolicy=rotate Valueattrs=(color=blue
weight=bold size=10);
yaxis grid minor;
```

Supporting Statements

Legend: Impacts the legend border, entries, legend title, inside/outside graph

Reference lines: Add horizontal or vertical

NEW SAS 9.4 M1

The DATASKIN= option has been added to the following plot types: DOT, DROPLINE, HBOX, VBOX, HLINE, VLINE, HIGHLOW, REFLINE, NEEDLE, SERIES, STEP, and VECTOR

Supporting Statements

ODS Graphics: Manage the settings for your graphics output, layout, sizing, adding tips

Example:

```
ods graphics on / reset width=3in
height=5in outputfmt=gif imagemap=on
imagename="MyBoxplot"
border=off;
```

Resources

Examples of ODS Graphics examples:

https://blogs.sas.com/content/graphicallyspeaking/

Collection of sample graphs by Procedure:

http://support.sas.com/sassamples/graphgallery/index.html

ODS Community:

https://communities.sas.com/t5/SAS-GRAPH-and-ODS-Graphics/bd-p/sas_graph

Tip Sheet

https://support.sas.com/rnd/app/ODSGraphics/TipSheet_ODSGraphics.pdf

……AND MANY MORE

Questions?

Thank You

MARY.HARDING@SAS.COM

sas.com